

# Coq/SSReflect の論理

## 1 Coq の論理

Coq は Calculus of Inductive Constructions (CIC) という型理論に基づいている。型理論は型付き λ 計算という計算体系と直観主義論理の証明論の間の同型 (カリー・ハワード対応) に基いた論理体系である。CIC は特に表現力が高い (一階述語論理や高階論理を含んでいる) が、まずその命題論理の部分から見ていくことにする

### 1.1 直観主義の命題論理

論理式 論理式は以下の結合子から定義される。

$$\begin{array}{ll}
 P, Q ::= & True \mid False \quad \text{定数} \\
 & | A \quad \text{論理変数} \\
 & | P \rightarrow Q \quad \text{含意} \\
 & | P \wedge Q \quad \text{論理積} \\
 & | P \vee Q \quad \text{論理和}
 \end{array}$$

否定はないが、便宜のために  $\neg P = P \rightarrow False$  とおく。

導出規則 自然演繹体系では真の論理式は以下の規則より導出される。

$\Delta$  を論理式の集合とする。True は常に  $\Delta$  に含まれる。

公理	$\Delta \vdash P$ ( $P$ は $\Delta$ に含まれる)	$\wedge$ 導入	$\frac{\Delta \vdash P \quad \Delta \vdash Q}{\Delta \vdash P \wedge Q}$
$\rightarrow$ 導入	$\frac{\Delta, P \vdash Q}{\Delta \vdash P \rightarrow Q}$	$\wedge$ 除去	$\frac{\Delta \vdash P \wedge Q \quad \Delta \vdash P \wedge Q}{\Delta \vdash P \quad \Delta \vdash Q}$
$\rightarrow$ 除去	$\frac{\Delta \vdash P \quad \Delta \vdash P \rightarrow Q}{\Delta \vdash Q}$	$\vee$ 導入	$\frac{\Delta \vdash P \quad \Delta \vdash Q}{\Delta \vdash P \vee Q \quad \Delta \vdash P \vee Q}$
矛盾	$\frac{\Delta \vdash False}{\Delta \vdash P}$	$\vee$ 除去	$\frac{\Delta \vdash P \vee Q \quad \Delta, P \vdash R \quad \Delta, Q \vdash R}{\Delta \vdash R}$

この導出規則から背理法や排中律が導けない。もしも古典論理の体系が欲しければ、矛盾の代わりに以下の規則のどちらかを使う。

$$\begin{array}{ll}
 \text{背理法} & \frac{\Delta, \neg P \vdash False}{\Delta \vdash P} \\
 \neg\neg\text{除去} & \Delta \vdash \neg\neg P \rightarrow P
 \end{array}$$

恒真式 命題論理の恒真式は True だけを仮定して導出できる式である。

例えば、 $P \rightarrow P \wedge P$  や  $P \rightarrow (P \rightarrow Q) \rightarrow Q$  は恒真式である。それぞれの導出を以下に示す。

$  \frac{\frac{P \wedge Q \vdash P \wedge Q}{P \wedge Q \vdash Q} \quad \frac{P \wedge Q \vdash P \wedge Q}{P \wedge Q \vdash P}}{P \wedge Q \vdash Q \wedge P} \quad (\wedge\text{導入})  $ $  \frac{P \wedge Q \vdash Q \wedge P}{\vdash P \wedge Q \rightarrow Q \wedge P} \quad (\rightarrow\text{導入})  $	$  \frac{P, P \rightarrow Q \vdash P \quad P, P \rightarrow Q \vdash P \rightarrow Q}{P, P \rightarrow Q \vdash Q} \quad (\rightarrow\text{除去})  $ $  \frac{P, P \rightarrow Q \vdash Q}{P \vdash (P \rightarrow Q) \rightarrow Q} \quad (\rightarrow\text{導入})  $ $  \frac{P \vdash (P \rightarrow Q) \rightarrow Q}{\vdash P \rightarrow (P \rightarrow Q) \rightarrow Q} \quad (\rightarrow\text{導入})  $
---	---

## 1.2 Coqでの証明

**変数宣言** まずは、準備として論理変数の宣言を行う。Section というコマンドを使うと、局所的な論理変数が宣言できるようになる。宣言自体は Variables コマンドを使う。そして、宣言範囲が終ると End コマンドでセクションを閉じる。Coq の出力をイタリック体で表示している。

```
Section Koushin.  
  
Variables P Q : Prop.  
P is assumed  
Q is assumed
```

変数を宣言するときはその型も書かなければならないが、論理式の型は Prop になる。

### 命題と証明

まず、二つ目の恒真式を証明してみよう。定理を宣言すると証明モードに移る。

```
Theorem modus_ponens : P -> (P -> Q) -> Q.      (* 名前を付けなければならない *)  
1 subgoal                                       (* 証明の状況が表示される *)  
  P, Q : Prop  
  =====  
  P -> (P -> Q) -> Q
```

線の上は  $\Delta$  の中身、下は結論だが、今は  $\Delta$  に変数の宣言しかない。P と Q が命題であるという宣言であり、P や Q が成り立つという仮定ではない。→ 導入にあたる Coq のタクティック (証明命令) は intros である。→ 除去は apply。公理は assumption。

```
Proof.  
  intros p pq.                                (* 仮定に名前を付ける (導入) *)  
1 subgoal  
  p : P  
  pq : P -> Q  
  =====  
  Q  
  
  apply pq.                                  (* 目標を仮定 pq の帰結とみなす (除去) *)  
1 subgoal  
  p : P  
  pq : P -> Q  
  =====  
  P  
  
  assumption.                                (* (公理) *)  
  Proof completed.  
Qed.  
modus_ponens is defined
```

実際の証明をもう一度みよう。

```
Theorem modus_ponens : P -> (P -> Q) -> Q.  
Proof.  
  intros p pq.  
  apply pq.  
  assumption.  
Qed.
```

証明が長い訳ではないが、少し違和感がある。直感的に二つ目の前提を一つ目に適用すれば結論が出るはずなのに、逆にそれを結論に「適用」している。証明を結論から作っていくので、考え方が逆になってしまい、証明が長くなると追いかたがたい。

## 2 SSReflect での証明

上の自然演繹体系は Coq の中の論理によく合っているが、証明の構築には不便である。証明の構築は、証明したい命題と仮定を下に書き、証明木を延して行く形で行うが、自然演繹の場合では、規則は結論 (右側) に対するものばかりで、その結論を変形させる「逆算」で行うことになる。通常の数学の証明は仮定から性質を導く「前進」方向で行うことが多い。Gentzen のシークエント計算が結論を扱う「右規則」とは別に仮定を扱う「左規則」も持っている。ここでは、そういう右規則を自然演繹に追加する形で SSREFLECT の証明方法を反映した論理体系を作る。

### 2.1 SSReflect の論理

新しい判定の形は

$$\Delta; P \vdash Q$$

$\Delta$  には名前付きの仮定が入っている。  $P$  は焦点と言い、左規則の対象となる。焦点は空白でもいい。  $Q$  は結論といい、右規則の対象である。

<b>Focus</b>	$\frac{\Delta; P \vdash Q}{\Delta; \vdash P \rightarrow Q}$	<b>公理</b>	$\Delta; P \vdash P$
<b>Unfocus</b>	$\frac{\Delta; \vdash P \rightarrow Q}{\Delta; P \vdash Q}$	<b>矛盾</b>	$\Delta; False \vdash P$
<b>Intro</b>	$\frac{\Delta, h : P; \vdash Q}{\Delta; P \vdash Q}$	<b>カット</b>	$\frac{\Delta; \vdash P \quad \Delta; P \vdash Q}{\Delta; \vdash Q}$
<b>Revert</b>	$\frac{\Delta; P \vdash Q}{\Delta, h : P; \vdash Q}$	<b><math>\wedge</math>-R</b>	$\frac{\Delta; \vdash P \quad \Delta; \vdash Q}{\Delta; \vdash P \wedge Q}$
<b>Apply-R</b>	$\frac{\Delta; \vdash P_1 \quad \dots \quad \Delta; \vdash P_n}{\Delta; P_1 \rightarrow \dots \rightarrow P_n \rightarrow Q \vdash Q}$	<b><math>\wedge</math>-L</b>	$\frac{\Delta; P \vdash Q \rightarrow R}{\Delta; P \wedge Q \vdash R}$
<b>Apply-L1</b>	$\frac{\Delta; \vdash P \rightarrow Q \quad \Delta; Q \vdash R}{\Delta; P \vdash R}$	<b><math>\vee</math>-R</b>	$\frac{\Delta; \vdash P \quad \Delta; \vdash Q}{\Delta; \vdash P \vee Q \quad \Delta; \vdash P \vee Q}$
<b>Apply-L2</b>	$\frac{\Delta; \vdash P \quad \Delta; Q \vdash R}{\Delta; P \rightarrow Q \vdash R}$	<b><math>\vee</math>-L</b>	$\frac{\Delta; P \vdash R \quad \Delta; Q \vdash R}{\Delta; P \vee Q \vdash R}$

規則は下から上へと読む。 **Focus** は結論の前提を焦点に移す。 **Unfocus** は逆に焦点された命題を結論の前提に戻す。 SSREFLECT ではこの二つの規則は暗に使われ、焦点と結論の前提が区別されない。 **Intro** は焦点に名前を付けて仮定に移す。 **Revert** は仮定を焦点に戻す。基本的にはこの4つの規則が仮定を動かしているだけだ。 **Apply-R** は焦点を使って結論を証明するが、その際焦点の全ての前提を証明する必要が新たに生じる。 **Apply-L1** は補題  $P \rightarrow Q$  を使って焦点を変える。 **Apply-L2** は逆に補題  $P$  を使って焦点の前提をみだす。この3つの規則を右側のカットで代用することができるので、論理的には必要ではないが、証明を自然に進めるのに使う。右側の規則は導入が自然演繹と変わらないが、右規則やカットが焦点を利用している。特に  $\wedge$ -R では焦点に  $P$  しか入らないので、  $Q$  が前提に戻される。

### 2.2 SSReflect のタクティック

SSREFLECT の論理は多くの規則をもっているが、Coq と違い各規則が独立したタクティックに対応している訳ではない。ほとんどのものが「move+修飾子」という形で書ける。

もう一回 `modus_ponens` を証明する。

```

Reset modus_ponens. (* module_ponens 以降の定義を忘れる *)
Require Import ssreflect. (* ssreflect のタクティック言語を使う *)

Theorem modus_ponens : P -> (P -> Q) -> Q.
Proof.
  move=> p. (* Intro *)
1 subgoal
  p : P
  =====
  (P -> Q) -> Q

  move/(_ p). (* Apply-L2 *)
1 subgoal
  p : P
  =====
  Q -> Q

  done. (* 公理や矛盾 *)
No more subgoals. (* 証明完了 *)
Restart. (* 証明をやりなおす *)
  by move => p /(_ p). (* 全てを一行にまとめる *)
Qed.

```

move の => と / は一緒にまとめられ、さらに by を他のタクティックの前に書くと、後に done を加えたときと同じ意味になる。

もう一つの恒真も試してみよう。

```

Theorem and_comm : P /\ Q -> Q /\ P.
Proof.
  move=> [] p q. (* ^-L, Intro, Intro *)
  p : P
  q : Q
  =====
  Q /\ P

  split. (* ^-R *)
2 subgoals

  P, Q : Prop
  p : P
  q : Q
  =====
  Q

  subgoal 2 is:
  P

  done. done.
No more subgoals.

Restart.
  by move=> [p q]; split. (* 名前を中に書く, タクティックを「;」でつなぐ *)
Qed.

```

恒真色々 証明状態の表示が作戦を読みにくくするので、これ以降は省くことにする。自分で Coq の中で実行して、確認して下さい。

```

Theorem or_comm : P \/ Q -> Q \/ P.
Proof.

```

```

move=> [p|q].          (* (V-L; 証明木が分岐するとき, 縦棒「|」で分ける *)
  by right.           (* V-R2 *)
  by left.            (* V-R1 *)
Qed.

```

Theorem DeMorgan :  $\sim (P \vee Q) \rightarrow \sim P \wedge \sim Q$ .

Proof.

```

rewrite /not.          (* 定義を展開する *)
move=> npq.
split=> [p|q].         (* 全てのタクティックで「=>」が使える *)
  apply: npq.         (* (Apply-R) *)
  by left.
  apply: npq.
  by right.
Qed.
End Koushin.          (* Section を閉じる *)

```

論理和に関する De Morgan の法則が証明できた。しかし、双対的な法則  $(\neg(P \wedge Q) \supset \neg P \vee \neg Q)$  は直観主義論理ではなりたたない。二重否定の除去か排中律を仮定する必要がある。それらの同値性を証明した上でその定理を証明する。

Section Classic.

Definition Classic := forall P : Prop,  $\sim \sim P \rightarrow P$ . (\*  $\neg\neg$  除去の定義 \*)

Definition EM := forall P : Prop,  $P \vee \sim P$ . (\* 排中律の定義 \*)

Lemma Classic\_is\_EM : Classic  $\leftrightarrow$  EM. (\*  $\neg\neg$  除去と排中律が同値 \*)

Proof.

```

rewrite /Classic /EM. (* 定義の展開 *)
split => [classic | em] P. (* A  $\leftrightarrow$  B := A  $\rightarrow$  B  $\wedge$  B  $\rightarrow$  A *)
  + apply: (classic) => nEM. (* classic を仮定から消さずに焦点におく *)
  have p : P. (* (カット) *)
    apply: classic => np.
    apply: nEM. by right.
    apply: nEM. by left.
  + by move: (em P) => []. (* P についての排中律で場合分けをする *)
Qed.

```

Definition は項や命題を定義する。ここでは二重否定除去と排中律の言明を定義するのに使う。forall P : Prop はこの言明が任意の命題に対して使えることを表している。

証明の中の「+」(または「-」や「\*」) は場合分けの構造を分かりやすくするためを使う。一時的に他のゴールが見えなくなる。

Variables P Q : Prop.

Theorem DeMorgan' : Classic  $\rightarrow \sim (P \wedge Q) \rightarrow \sim P \vee \sim Q$ .

Proof.

```

move=> /Classic_is_EM em npq. (* Apply-L1, Intro, Intro *)
move: (em P) => [p|np].      (* 排中律で場合分け *)
+ move: (em Q) => [q|nq].
  - elim: npq. (* 矛盾 *)
    by split.
  - by right.
+ by left.
Qed.
End Koushin. (* Section を閉じる *)
Check DeMorgan'. (* 命題の言明を確認する *)
DeMorgan'

```

: forall P Q : Prop, Classic  $\rightarrow \sim (P \wedge Q) \rightarrow \sim P \vee \sim Q$

Section から出ると、使われた命題変数が各定理の言明に自動で追加される。

### 論理規則と tactic の対応

論理規則	作戦	論理規則	作戦
Intro	move => <i>h</i>	公理	done, by
Revert	move: <i>h</i>	矛盾	elim, suff: False
Apply-R	apply	$\wedge$ -R	split
Apply-L1	move/ <i>e</i>	$\wedge$ -L	move => []
Apply-L2	move/(_ <i>e</i> )	$\vee$ -R	left, right
カット	have, suff, move: ( <i>e</i> )	$\vee$ -L	move => [[]]

各タクティックの詳細な説明は次回 (述語論理) のときに述べる。

練習問題 2.1 以下の定理を Coq で証明せよ。

Section Coq1.

Variables P Q R : Prop.

Theorem imp\_trans : (P -> Q) -> (Q -> R) -> P -> R.

Theorem not\_false : ~False.

Theorem double\_neg : P -> ~~P.

Theorem contraposition : (P -> Q) -> ~Q -> ~P.

Theorem and\_assoc : P  $\wedge$  (Q  $\wedge$  R) -> (P  $\wedge$  Q)  $\wedge$  R.

Theorem and\_distr : P  $\wedge$  (Q  $\vee$  R) -> (P  $\wedge$  Q)  $\vee$  (P  $\wedge$  R).

Theorem absurd : P -> ~P -> Q.

Definition DM\_rev := forall P Q, ~ (~P  $\wedge$  ~Q) -> P  $\vee$  Q.

Theorem DM\_rev\_is\_EM : DM\_rev <-> EM.

End Coq1.